

Задание 9 старожил КЕГЭ: работа в электронных таблицах проверяет «умение обрабатывать числовую информацию в электронных таблицах. В спецификации сказано, что для выполнения задания – требуется использование специализированного программного обеспечения. Примерное время, отводимое на выполнение 4 минуты. Задание также базового уровня.

9	Умение обрабатывать числовую информацию в электронных таблицах	4.2	2.13	Б	да	1	6
---	--	-----	------	---	----	---	---

И также, как и с заданием 6, с этим справилось всего примерно 20%. Хотя нижняя граница процента выполнения заданий базового уровня 60%.

Давайте разбираться, что же вызвало такие затруднение у учащихся при выполнении этого задания в ЕГЭ 2023 года.

Но для начала посмотрим, какие задания могут быть в № 9 на КЕГЭ.

Можно немного сказать про самые популярные функции в электронных таблицах: МАКС, МИН, СРЗНАЧ, СУММА, ЕСЛИ, НАИБОЛЬШИЙ, ОСТАТ, СЧЕТЕСЛИ, ЛОГИЧЕСКИЕ И, ИЛИ и др.

Начнем с такого номера.

1 Тип 9 № 38588 📁 (1 балл) 🌟 ⚠️ ⓘ

Откройте файл электронной таблицы, содержащей в каждой строке три натуральных числа.

[9.xlsx](#)

Выясните, какое количество троек чисел может являться сторонами треугольника, то есть удовлетворяет неравенству треугольника. В ответе запишите только число.

В данном номере можно поступить по разному. Но основная идея состоит в том, чтобы расположить числа в порядке убывания и проверить неравенство треугольника (большая сторона должна быть меньше суммы двух других). Для расположения чисел в порядке убывания (возрастания) удобно пользоваться функцией НАИБОЛЬШИЙ(диапазон чисел; номер максимума)

Ответ: 2453

2 Тип 9 № 58321 📁 (1 балл) 🌟 ⚠️ ⓘ

Откройте файл электронной таблицы, содержащей в каждой строке четыре натуральных числа.

[Задание 9](#)

Определите количество строк таблицы, содержащих числа, для которых выполнены оба условия:
 — наименьшее из четырёх чисел более чем в шесть раз меньше суммы трёх других;
 — произведение наибольшего и наименьшего числа больше произведения оставшихся чисел.
 В ответе запишите только число.

Решение: ОДИН ИЗ ВАРИАНТОВ: расположим числа в порядке убывания с помощью функции НАИБОЛЬШИЙ.

Потом проверим оба условия с помощью функции ЕСЛИ. И потом с помощью логического умножения (подходит обычное умножение)

3

Тип 9 № 52180 (1 балл)

В каждой строке электронной таблицы записаны пять натуральных чисел.

Определите, сколько в таблице строк, для которых выполнены следующие условия:

- все числа в строке различны;
- чётных чисел больше, чем нечётных;
- сумма чётных чисел меньше суммы нечётных.

В ответе запишите число — количество строк, для которых выполнены эти условия.

Ответ: 241

Решение (с сайта РЕШУ ЕГЭ):

Сначала будем проверять сколько раз в строке встречается число. Для этого в ячейку F1 запишем формулу

=СЧЁТЕСЛИ(\$A1:\$E1;A1)

и скопируем её во все ячейки диапазона **F1:J6400**.

В ячейке K1 запишем формулу **=СУММ(F1:J1)**. Если число в ячейке равно 5, то в ряду все числа встречаются один раз, этим мы проверим первое условие отбора строк.

Скопируем формулу во все ячейки диапазона **K1:K6400**.

В ячейку L1 запишем формулу, которая даст 1, в случае если число в ячейке A1 нечетное, и 0 если число четное:

=ОСТАТ(A1;2)

и скопируем её во все ячейки диапазона **L1:P1**.

В ячейке Q1 посчитаем сумму ячеек диапазона **K1:K6400**, для этого введем формулу

=СУММ(L1:P1)

Если сумма меньше 3, то в строке больше четных чисел. Этим мы проверим второе условие отбора строк. Скопируем её во все ячейки диапазона **Q1:Q6400**.

В ячейке R1 посчитаем сумму четных цифр строки. Для этого запишем формулу:

=СУММЕСЛИ(L1:P1;0;A1:E1)

и скопируем её во все ячейки диапазона **R1:R6400**.

В ячейке S1 посчитаем сумму нечетных цифр строки. Для этого запишем формулу:

=СУММЕСЛИ(L1:P1;1;A1:E1)

и скопируем её во все ячейки диапазона **S1:S6400**.

В ячейке T1 будем получать 1, если сумма чётных чисел меньше суммы нечётных, и 0 в обратном случае. Для этого запишем формулу:

=ЕСЛИ(S1>R1;1;0)

и скопируем её во все ячейки диапазона **T1:T6400**.

В ячейке U1 будем получать 1, если строка удовлетворяет всем условиям задачи, и 0 если не выполнены условия. Для этого запишем формулу:

=ЕСЛИ(И(T1=1;K1=5;Q1<3);1;0)

и скопируем её во все ячейки диапазона **U1:U6400**.

Окончательно, с помощью формулы **=СУММ(U1:U6400)** получим ответ — 241.

Есть другие способы решения этого номера, но все они достаточны громоздки для базового уровня.

Рассмотрим номер с прошлого года ЕГЭ. И задания ЕГЭ 2023 года были еще сложнее.

Требовалось большое количество вычислений.

4 Тип 9 № 59833 (1 балл)

Откройте файл электронной таблицы, содержащей в каждой строке шесть натуральных чисел. Определите количество строк таблицы, содержащих числа, для которых выполнены оба условия:
— в строке есть только два равных числа, остальные 4 различны;
— среднее арифметическое повторяющихся чисел меньше чем среднее арифметическое остальных чисел строки.
В ответе запишите только число.

Задание 9

Источник: ЕГЭ по информатике 20.06.2023. Основная волна, Дальний Восток

Решение:

Решение.

Для решения задачи выпишем в столбцы от G до L повторяющиеся значения. Выпишем в ячейку G1 формулу

=ЕСЛИ(СЧЁТЕСЛИ(\$A1:\$F1;A1)=2;A1;"").

Данная формула вернет в ячейку значение из ячейки A1 если оно встречается два раза. Скопируем формулу на диапазон G1:L30000. В ячейку M1 выпишем формулу

=ЕСЛИ(СЧЁТЕСЛИ(G1:L1;">0")=2;СУММ(G1:L1)/2;0)

и скопируем ее на диапазон M1:M30000. В данном столбце будет среднее арифметическое чисел, которое повторяется в строке 2 раза или 0, если таких значений нет. В ячейку N1 выпишем формулу

=ЕСЛИ(СЧЁТЕСЛИ(\$A1:\$G1;A1)=1;A1;"")

и скопируем ее на диапазон N1:S30000. Данная формула вернет значение в ячейки, если они не повторяются. В ячейку T1 выпишем формулу

=ЕСЛИ(СЧЁТЕСЛИ(N1:S1;">0")=4;СРЗНАЧ(N1:S1);0)

и скопируем ее на диапазон T1:T30000. В данном столбце будет среднее арифметическое всех чисел строки, если строка удовлетворяет условию задачи. В столбце U будем считать количество строк, удовлетворяющих нашему решению. Для этого введем формулу

=ЕСЛИ(M1 < T1;1;0)

и скопируем ее на весь диапазон U1:U30000. Посчитаем количество строк, удовлетворяющих нашему условию формулой: **=СУММ(U1:U30000).**

Ответ: 1483

Что же следует ожидать в №9 на ЕГЭ 2024 года? Скорее всего по-прежнему будут достаточно сложные номера, где нужно будет проверить три условия например. Или нужно будет посчитать количество строк, где ХОТЯ бы одно условие выполнялось.